Mth 351 Least Squares

Mth 351 Fall 2001 Assignment 2. Due Oct 29.

Oct 22 2001 Maple 5 and 6

Bent E. Petersen

Filename: 351f2001-assign-02-ls.mws

```
> restart;
```

A thermistor is a device which can be used as a temperature sensor since its resistance to an electrical current changes with temperature. It is necessary first to calibrate it. For a thermistor the theoretical relation between the absolute temperature T and the resistance R is given by

```
> R=A*exp(B/T); 1/T = a+b*log(R);
```

$$R = A\,\mathbf{e}^{\left(\frac{B}{T}\right)}$$

$$\frac{1}{T} = a + b\,\ln(R)$$

for certain constants a and b. In practice the relation is not so tidy. The equation used in practice is the empirical Steinhart-Hart equation

```
> 1/T=a+b*log(R)+d*(log(R))^3;
```

$$\frac{1}{T} = a + b\,\ln(R) + d\,\ln(R)^3$$

where R is measured in Ohms. Now both R and T have a distinguished 0 point so we do not need to worry about translation invariance, but it is hard to believe that ohms have some special physical significance. In other words the form of the equation ought to be invariant under a change of units. Yet changing to kilo-ohms would introduce a log(R)^2 term. Indeed, to achieve scale invariance we need to add a log(R)^2 term from the beginning.

In general we are not in posession ot the actual resistance of the thermistor, but just some value proportional to it. Thus the scale invariance is crucial.

If we do so, we have

```
> eqn1:=S=a+b*log(R)+c*(log(R))^2+d*(log(R))^3;
```

$$eqn1 := S = a + b\,\ln(R) + c\,\ln(R)^2 + d\,\ln(R)^3$$

Here we let S = 1/T so we will have a linear relation to fit (linear in powers of log(R)).

Problem: Your assignment is to fit this equation (by the method of least squares) to some gritty measurements of actual themistors generously provided by Erik Petersen. You should also examine c to see if it is relatively small as would be required by the Steinhart-Hart model. You should also do a least squares fit for the Steinhart-Hart model (even if it doesn't make sense for our data) and graphically compare the results.

If you elect to do the problem in Maple, everything you need is provided below - you just have to sort it out. If you elect to do it some other way, you are on your own.

The original data consists of an ADC value (voltage across the thermistor as measured by an analog to digital converter) and the temperature in degrees Celsius. We add a bias of 32768 to the ADC measurement to obtain a value proportional to the actual voltage drop across the thermistor, which in turn is proportional to the resistance. The idea is to provide the constants a, b, c, d for each thermistor so it can be used to determine temperatures directly from the ADC values.

The temperature will also have to be adjusted by adding 273.15 to the Celcius reading to convert it to absolute Kelvin.

```
> Sadj:=T->1/(T+273.15);
```

$$Sadj := T \to \frac{1}{T + 273.15}$$

```
> Radj:=ADC->evalf(ADC+32768);
```

$$Radj := ADC \to \mathrm{evalf}(ADC + 32768)$$

Note the evalf() here is critical. Without it Maple will retain the exact (symbolic) values of log(R) for as long as possible in computing the least squares fit.Whether you get an error message or a crash depends on the version of Maple, but you will not like what you get in either case.

The data sets are for two different thermistors

Data Set 1

```
> ADC1raw:=[-29292,-25792,-22338,-18446,-15124,-11267,-7469,-3843];
```

$$ADC1raw := [-29292, -25792, -22338, -18446, -15124, -11267, -7469, -3843]$$

```
> T1raw:=[2.426,5.333,7.997,10.814,13.085,15.600,17.968,20.143];
```

$$T1raw := [2.426, 5.333, 7.997, 10.814, 13.085, 15.600, 17.968, 20.143]$$

```
> R1:=map(Radj,ADC1raw);
```

$$R1 := [3476., 6976., 10430., 14322., 17644., 21501., 25299., 28925.]$$

```
> S1:=map(Sadj,T1raw);
```

$S1$:= [.003628763027, .003590883465, .003556858156, .003521573157, .003493632854,

.003463203463, .003435033217, .003409559724]

Data Set 2

```
> ADC2raw:=[-29564,-26068,-22618,-18730,-15408,-11553,-7755,-4127];
```

ADC2raw := [-29564, -26068, -22618, -18730, -15408, -11553, -7755, -4127]

```
> T2raw:=[2.426,5.333,7.997,10.814,13.085,15.600,17.968,20.143];
```

T2raw := [2.426, 5.333, 7.997, 10.814, 13.085, 15.600, 17.968, 20.143]

```
> R2:=map(Radj,ADC2raw);
```

R2 := [3204., 6700., 10150., 14038., 17360., 21215., 25013., 28641.]

```
> S2:=map(Sadj,T2raw);
```

S2 := [.003628763027, .003590883465, .003556858156, .003521573157, .003493632854, .003463203463, .003435033217, .003409559724]

The most convenient facility for doing least squares in Maple is in the stats[fit] package.

```
> with(stats[fit]):with(stats[statplots]):with(plots):
Warning, the name changecoords has been redefined
```

I threw in some plotting packages too in case you want to try some sophisticated plotting.

```
> lsq:=(Rdata,Sdata)->leastsquare[[R,S],eqn1,{a,b,c,d}]([Rdata,Sdata
  ]):
```

Our function lsq() returns an equation in S and R. We would probably prefer a relation between the ADC values and the temperature in Celsius.

```
> eqn_data1:=lsq(R1,S1):
> TC1:=ADC->-273.15+1/subs(R=ADC+32768.0,rhs(eqn_data1)):
```

Replace the colons by semicolons above to see the results.

At this point you may wish to plot TC1(ADC) for ADC from about -30000 to -4000. If you a calibrating the thermistors for a customer you would probably give the customer a graph and also the values of the parameters a, b, c and d.

Now repeat the analysis for the second data set. Also fit the original Steinhart-Hart model. How well does your least squares fit actually fit the original data? Discuss.

```
>
```